
Abstract

Wishbone compliant smart Pulse-Width  Modulation
(PWM) IP

Magnus Norgren

ÅAC Microtec AB in Uppsala, Sweden develops miniaturized and robust
multifunctional electronics systems for space and aerospace. Since much of the
hardware implementations requires custom designs or its functions to easily be
changed, it is a great benefit to have in-house developed so called Intellectual
Property cores (IP cores) that can be adapted to new conditions when necessary. An
IP core describes digital functions and consists of chunks of code written in a
Hardware Description Language (HDL) such as VHDL or Verilog. These IP cores can
be combined to form desired functionality and then implemented in silicon such as an
ASIC or on programmable logic such as a FPGA. Since ÅAC products may include
control of electrical motors, an IP core with motor control functionality based on the
Texas Instruments TMS320x281x Event Manager circuitry was designed and
implemented in Verilog. Further, in order to be able to on-chip connect the Event
Manager to other IP cores such as the OpenRISC soft CPU, an interface of type
Wishbone System-on-Chip (Soc) Interconnection Architecture was mandatory. After
successful simulations, the Event Manager IP core was implemented together with the
OpenRISC on ÅAC FPGA development board and tests of all Event Manager
functionality were conducted and verified. The different parts of this thesis work are
design of Event Manager IP core, design of Wishbone interface, implementation on
development board together with OpenRISC soft CPU, implementation of test
software for the Event Manager IP core to be executed on the OpenRISC and the
tests themselves.

 1 Introduction

This report describes the thesis work of designing an IP core1 with Pulse-Width modulation 
capabilities for ÅAC Microtec AB in Uppsala. 

 1.1 Background

The TMS320x281x Event Manager[1] is a part of a DSP2 series available from Texas Instruments 
which purpose is to handle different events such as capture, compare & PWM and Quadrature-
Encoder Pulse. These events are mostly correlated to motor control applications.

ÅAC Microtec AB in Uppsala is developing miniaturized and robust multifunctional electronics 
systems for space and aerospace and some of the targets may include controlling of brushed or 
brushless DC motors. Since ÅAC already have a SoC3 platform based on OpenRISC[2] architecture 
running on FPGA, an IP Core based on TMS320x281x Event Manager would would fill the need 
for motor control abilities.

 1.2 Hardware architecture

Target architecture for developing and testing the IP Core was done on ÅAC development platform 
for their μRTU4 that includes all necessary powering and I/O pins needed. This in turn has a 
development board[3] from ORSoC5 mounted on top that has an Actel ProASIC 3E FPGA[4] as 
programmable logic device. It is running a soft CPU core based on OpenRISC 1200 of which the 
Event Manager IP Core should be connected to.

 1.3 Mission brief

• Design an IP core with functionality based on TMS320x281x Event Manager

• Design an Wishbone 8-bit interface for the Event Manager IP Core

• Implement Event Manager design together with OpenRISC on development board

• Test all Event Manager functions

1 Intellectual Property Core, section 2.3 Field-Programmable gate array
2 Digital Signal Processor
3 System-on-Chip
4 μRTU – micro Remote Terminal Unit
5 An electronics company located in Stockholm with focus on SoC solutions

Figure 1: μRTU Development Board: μRTU Development Board



 2 Theory

This chapter describes the theory of the most important functions of the Event Manager IP core and 
includes an introduction to FPGA, IP cores and OpenRISC.

 2.1 Pulse-Width Modulation

Pulse-Width modulation is commonly used in industrial applications for electrical motor control but 
is also used in audio applications, such as compact class-D power amplifiers, and drivers for light 
sources based on light-emitting diodes.

 2.1.1 Principle

Pulse-width modulation or PWM is a signaling technique in which a periodic signal only can take 
one of two discreet levels but having a variable active time during a fixed period thus giving a pulse 
of variable width. Hence the naming Pulse-Width modulation.

Duty cycle is a measure of how long the signal is active and corresponds to a percentage of the 
PWM period as shown in Figure 2 and equation (1) below. The inverse of the PWM period is the 
PWM switching frequency or just PWM frequency (2).

DutyCyclePWM =
Active part

PWM period
⋅ 100 % (1)

f PWM =
1

PWM period
(2)

From this point of view the Duty Cycle can be seen as intermediate value between on and off that is 
to be transferred either as power or as signal value. If for instance the PWM signal is being passed 
through a RC low-pass filter with cutoff frequency much lower than the PWM frequency the output 
will be a DC voltage. The DC voltage corresponds to the Duty Cycle percentage of the PWM active 
voltage as shown in Figure 3. This is an idealized example and in reality the output will have a 

Figure 2: Duty Cycle of PWM signal



ripple that has the same frequency as the PWM switching frequency and will be more significant if 
the PWM frequency can not be made high enough compared to the cutoff frequency. On the other 
hand if that can be done, a very good approximation of a stable DC output can be achieved.

 

 2.1.2 Electric motor control

A very common method for controlling brushed DC motors is by using a so called H-bridge. A H-
bridge is an electronic circuit with shape of the letter “H” in which each leg has an on/off switch 
and the the topmost legs are connected to a power source whereas the ones in the bottom are 
connected to ground. The horizontal bar consists of the brushed DC motor.

By manipulating the switches in different ways, and thus the current flow, the motor can be set to 
free running, rotate in positive (or forward) direction, rotate in negative (or backward) direction or 
breaking as shown in Figure 4.

In practice, the switch elements are some type of transistors and manipulation is been done by 
PWM signals. Special care most be done so that both legs on one side does not conduct at the same 
time since that will cause a short circuit and most probable will damage the circuit.

Since a DC motor has inductive and resistive characteristic it will act as a LP filter and by tuning 
the PWM switching frequency, a steady DC current can be achieved. A to low PWM frequency will 
yield a jerky drive but on the other hand, a to high PWM frequency will create unnecessary power 
losses in the switching transistors.

4

Figure 3: PWM signal passed throw a LP-filter or load with LP-filter characteristic



The benefit of using a H-bridge is that drive in both directions is possible. If drive in only one 
direction is sufficient other solutions such as a half-bridge can be used. A half-bridge is basically a 
half H-bridge where the legs on one side is removed and the motor is directly connected to power as 
in Figure 5.

Figure 5: Half-bridge configuration

Figure 4: Current flow through DC motor depending on switch settings in an H-bride



 2.1.3 PWM generation

There exists many different methods for generating PWM signals. One method commonly used in 
MCU:s6 is by using a timer/counter register together with a compare register.

In such configuration the counter register will “ramp up” its value in increments of 1 and when the 
values of the counter and compare registers match, the PWM signal will enter its active phase. The 
counting range is either up to when the register overflows or when it has reach a certain 
configurable value. Either way it will restart from zero and the PWM signal ends its active phase.  
The counting range and the frequency of the increments defines the period of the timer/counter and 
thus also defines the PWM period. It is easy to understand that the Duty Cycle depends on the timer 
counting range AND the compare value and is independent of the actual counting frequency. The 
counting range of the timer is what defines the accuracy of the PWM signal since more steps will 
yield finer granularity. This stands in conflict with high PWM frequency because in order to have 
both, a fast timer counter is required and a compromise between the two might be necessary.

In this method the leading edge of the PWM signal varies with the compare value and the trailing 
edge is fixed to when the timer is zero. Sometimes the opposite might be wanted to obtain that it is 
just a matter of inverting the PWM signal. This also has the advantage that the Duty Cycle will be 
directly proportional to the compare value.

Figure 6 visualizes the timer ramp and PWM generation for 2 different compare-values and also the 
inverted PWM signal. Here the timer range is 8 (0 – 7) and 1:st compare-value is 5 and the 2:nd 
compare value is 2. Note that in practice the actual counting range is usually much larger using up 
to 8-bit (256 values) or 16-bit (65536 values).

With this reasoning and using (1) and (2) the following relations for calculating the Duty Cycle (3), 
and the inverse variant (4), can be derived where TMRtop corresponds to maximum timer counter 
value and CMPR corresponds to active compare value.

6  MicroController Unit

Figure 6: PWM-(and its inverse )-generation by timer and compare unit



DutyCycle PWM = (TMR top+1 − CMPR

TMRtop+1 ) ⋅ 100 % = (1 −
CMPR

TMR top+1 ) ⋅ 100 % (3)

DutyCycle PWM = 100 % − DutyCyclePWM =
CMPR

TMRtop+1
⋅ 100% (4)

 2.2 Quadrature-Encoder Pulse

Quadrature-encoder pulse is a digital method to measure direction and speed of a rotating device. 
This is done by on the rotating device generate two pulse trains which has a 90 degree phase 
difference. Since the pulse frequency, which is the same for both pulse trains, corresponds to 
rotational speed and by comparing the edges and levels of the two pulse trains, rotational direction 
can be determined as shown in Figure 7.

 

 2.3 Field-Programmable Gate Array

Field-Programmable Gate Array or FPGA is a reprogrammable digital device in which arbitrary 
combinatorial and sequential logic functions can be implemented. Development is done by using a 
so called Hardware Description Language or HDL where VHDL7 and Verilog are the most 
commonly used ones. In HDL both sequential and combinatorial (parallel) statements can be used 
and is what foremost separates HDL from conventional programming languages. A way of sharing 
HDL code is the use of IP cores8 which is blocks of code that describes a certain functionality such 
as memory controllers, communication devices or even CPUs such as the OpenRISC 1200. In this 
way IP cores can be viewed as building blocks in a digital design. The IP cores are either shared 
under an open license such as from OpenCores or proprietary licensed from a commercial vendor 

7 Very high speed integrated circuit Hardware Description Language
8 Intellectual-Property Core

7

Figure 7: Rotational direction of quadrature pulse trains



such as Altera9 or Microsemi10. 

A FPGA has many benefits compared to an ASIC11 such as:

• Rapid prototyping - the FPGA can be programmed, tested – reprogrammed, 
tested again, etc

• In field programmable - bugs can be corrected after the FPGA been mounted

• In field upgradeable - added functionality can be implemented after the FPGA 
has been mounted in design/product

On the opposite side is a FPGA usually slower and uses more power than an ASIC. 

Leading manufacturers of FPGA chips are Xilinx12, Altera and Microsemi (formely Actel).

 2.4 Wishbone Bus

There exists many different kinds of IP cores and in order to be able to easily incorporate these in a 
design a common interconnection scheme is desirable. One such method is the Wishbone Bus[5]. 
The Wishbone bus is a open source system-on-chip bus maintained by OpenCores and allows for 
many different forms of connection where master/slave configuration, Figure 8, is the most basic 
one. Other more complex connections such as Shared bus, Pipline and Cross bar switch are also 
supported.

Although the Wishbone bus incorporates many different kinds of signals not all are mandatory in a 
design but usually a slave device uses at least the following subset:

9 http://www.altera.com/
10 http://www.microsemi.com/
11 Application-Specific Integrated Circuit
12 http://www.xilinx.com/

Figure 8: Master/Slave Wishbone bus connection



WB slave Signal name Description

RST_I Reset signal, put device in known initial condition

CLK_I WB clock, bus synchronization clock

ADR_I Slave address, register or function selection

DAT_I Data in, data to slave, 8, 16, 32 or 64 bit wide

DAT_O Data out, data from slave, 8, 16, 32 or 64 bit wide

WE_I Write enable, on active strobe and cycle signals, write if set read otherwise

STB_I Strobe, slave bus cycle activation

CYC_I Cycle, active bus cycle

ACK_I Acknowledge, signal acknowledge of active bus operation to Master

Tabell 1: Common Wishbone signals

 2.5 OpenRISC 1200

OpenRISC is a CPU architecture of RISC13 type with HDL code written in Verilog. It has 32-bit 
wide registers and instruction set and has features like data/instruction caches and MMU14. 
OpenRISC in now included in the Linux mainline kernel and has support of the GCC15 suite.

The IP core is maintained by Opencores under an open source license and ORSoC has developed 
build scripts for total SoC solution based on OpenRISC that simplifies inclusion of peripheral 
modules or other customizations. All IP core interconnections are based on the Wishbone bus.

13 Reduced Instruction Set Computing
14 Memory Management Unit
15 GNU Compiler Collection

 3 Event Manager

This chapter presents how the Event Manager IP core and its parts are implemented.

 3.1 Event Manager Top Level

The Event Manager is made up of discrete function blocks which purpose foremost is motor and/or 
motion control. These blocks are:

• General purpose timers - Timing functionality

• Full Compare units - PWM generation

• Capture units - Timestamping of external events

• Quadrature-Encoder pulse units - Direction and speed measurement 

• CPU IRQ Event handler - CPU interrupt request signals

Figure 9: Top level of Event Manager



All registers are 16-bit wide and are resident in an internal 6-bit address space of the Event 
Manager. The registers are accessed and manipulated by the address bus, data-in/out buses and we 
signal. The address register map is presented in Appendix A1.

The event manager uses only active high logic levels except for PWM outputs of which the polarity 
can be controlled. If another active state is desired on external signals this must be configured at 
instantiation of Event Manager at its top level.

Event Manager IP-Core Signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

adr_i 6 in Register Selection

data_i 16 in Data input

we_i 1 in Enable register write

sys_clk_i 1 in System Clock 

ext_clk_i 1 in External timer counting trigger

reset_i 1 in Asynchronous reset of device

data_o 16 out Data output

ctrip_i  # [1 … 6] 1 in Compare trip

tdir_i # [1 ... 2] 1 in External timer counting direction

cap_i # [1 … 3] 1 in Capture trip

pwm_o # [1 … 12] 1 out PWM output – 6 pairs

ctrip_int_o # [1 … 6] 1 out Compare trip interrupt

cmpr_int_o # [1 … 6] 1 out Compare match interrupt

tpr_int_o # [1 … 2] 1 out Timer period match interrupt

tunderflow_int_o # [1 … 2] 1 out Timer underflow interrupt

toverflow_int_o # [1 … 2] 1 out Timer overflow interrupt

capture_event_int_o # [1 … 3] 1 out Capture event interrupt

adcsoc_o 1 out ADC Start-Of-Conversion

Table 2: Event Manager Signals

 3.2 Event Manager Modules

The hardware layout of the Event Manager is mostly by using discrete modules instead of using one 
or few integrated block(s). This approach makes it easier to change the layout if another hardware 
configuration and/or functionality is desired.

11



 3.2.1 General Purpose Timer

The general purpose timers, gptimer are the central part of the Event Manager since it is involved in 
all supported operations. By supporting different counting modes and events many kinds of timing 
situations can be covered.

The heart of the gptimer is the 16-bit wide counting register, TxCNT which holds present count. 
The counting interval is configured by the period register, TxPR which allow the counting period to 
span from 0x0001 to 0x10000. The period register by itself is doubled buffered and is reloaded 
when the counting register is underflow16, 0x0000. This allows for safe on-the-fly reconfigurations 
of the counting period.

A prescaler can be used to further configure the counter frequency and is configurable in steps by 
f/1, f/2, f/4 … f/128 where f is the frequency of tick source.

Supported counting modes are:

• up-counting mode: the counting value is cleared to 0x0000 (BOTTOM) after a Period match 
(TOP) or when counter reaches 0xFFFF (MAX). Timer counting will include the value of 
period register giving a counting period of TxPR + 1.

• up/down-counting mode: the counting direction is changed at the endpoints BOTTOM and 
TOP/MAX. Counting period is 2 * TxPR.

• directional up/down-counting mode: is supported by using the external direction pin, tdir_i 
where counting direction follows the value of tdir_i. The counting period will be same as in 
up-counting mode.

• QEP: when Quadrature-Encoder Pulse is used as counting mode both counter frequency and 
direction is dictated by the QEP circuitry. The counting period will be same as in up-
counting mode.

Besides for the system clock, sys_clk an external clock can be used to trigger the counter through 
the prescaler. Since it is sampled by an edge detector the external clock frequency must not be 
greater than the frequency of sys_clk. By the same analogy the built-in Quadrature-Encodeder Pulse 
unit, qep can be used to trigger the counting register as well as counting direction.

The gptimer module also support external start-of-timer signal, tenable_ext_i and external period 
register. This is useful when connecting two or more gptimer modules and synchronous support is 
needed.

16 The definitions for underflow and overflow in the Event Manager is when TxCNT IS 0x0000 resp 0xFFFF and
not changing from 0x0000 to 0xFFFF or vice versa.

GPTimer signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

tenable_ext_i 1 in External Timer Start trig

cnt_we_i 1 in Enable write of counting register

cnt_i 16 in Data to counting register

tdir_i 1 in External counting direction 

qep_dir_i 1 in Quadrature-encoder-pulse direction

ext_clk_i 1 in External clock trig

qep_clk_i 1 in Quadrature-encoder-pulse clock trig

tcon_we_i 1 in Enable write of Timer Control register

tcon_i 16 in Data to Timer Control register

pr_ext_i 16 in External Period compare

pr_we_i 1 in Enable write of Period register

pr_i 16 in Data to Period register

cnt_dir_o 1 out Timer counting direction

cnt_o 16 out Timer counting value

tcon_o 16 out Timer Control value

pr_match_o 1 out Period Match pulse

overflow_o 1 out Timer Overflow pulse (TxCNT at 0xFFFF)

underflow_o 1 out Timer Underflow pulse (TxCNT at 0x0000)

13

Figure 10: GPTimer



pr_o 16 out Period value

Table 3: GPTimer signals

 3.2.2 Compare Unit

The compare modules are fairly basic in construction but was still choosed to be created as a 
module. When enabled the output from the gptimer (or any other 16-bit value) is compared to the 
CMPR register. The compare register is doubled buffered with CMPR buf register and reload 
conditions can be configured by cld_i. The reload conditions are:

• reload on timer underflow,

• reload on timer underflow or timer period match

• reload immediately.

When a compare match occurs a pulse with length of the sys_clk_i period put out on cmpr_match_o 
and PWM signal will change polarity according to timer counting direction, cnt_dir_i. The PWM 
signal will always be cleared to zero on timer underflow.

It is designed to be used with an external configuration register, COMCONx which holds the 
configuration values of ecmpr_i and cld_i.

It should be noted that the PWM signal is NOT used in the Event Manager where a dedicated PWM 
unit is used in conjunction with the cmpr_match_o signal instead. The PWM capability of the 
compare unit is kept in case of situations were another hardware layout it is needed.

Compare unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

Figure 11: Compare Unit



cnt_i 16 in Timer counting value

ecmpr_i 1 in Enable compare operation

pr_match_i 1 in Timer Period match

underflow_i 1 in Timer Underflow

cnt_dir_i 1 in Timer counting direction

cld_i 2 in Compare reload condition

we_i 1 in Enable write of compare buffer register

cmpr_i 16 in Data to Compare buffer register

cmpr_match_o 1 out Compare match pulse

pwm_o 1 out PWM output

cmpr_o 16 out Compare value

Table 4: Compare unit signals

 3.2.3 Compare Match Interrupt Hold Unit

Since the Compare units can deliver Compare-Match events in a rapid pace and it might be 
unnecessary for software to have to take care of every one of the interrupts these produces, a 
Compare Match Interrupt Hold unit was created. This unit is located between the Compare Unit and 
the Interrupt Generator. It is composed of a configurable compare register, cmpr_reg and a counting 
register, cnt_reg that up-counts it value on every Compare Match event and when the two registers 
match the Compare Match event is passed to the Interrupt Generator. cnt_reg is cleared and the 
process can restart. A cmpr_reg value of zero will pass every Compare Match event.

Compare Match Interrupt Hold unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

Figure 12: Compare Match Interrupt Hold unit



reset_i 1 in Asynchronous reset

cmpr_match_i 1 in Compare Match pulse

we_i 1 in Enable write of cmpr_reg 

cnt_load_i 16 in Count value

cnt_o 16 out Counted value

cmpr_match_o 1 out Pass cmpr_match_i when cnt_reg = cmpr_reg

Table 5: Compare Match Interrupt Hold unit signals

 3.2.4 PWM generation Unit

The PWM generation unit supports ordinary PWM generation from Compare Match events but it 
also supports generation of Space Vector PWM as described in TI TMS320x281x documentation. 
The Space Vector PWM generation requires event signals from two Compare units which yields in 
three PWN signals. Therefore three Compare units can be attached to the PWM generation unit and 
in ordinary PWM mode each Compare Match event generates its own PWM signal whereas in 
Space Vector PWM mode the third Compare Unit will not be able to generate a PWM signal 
(although it can still generate Compare Match events for interrupt generation) 

PWM generation unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

cmpr_match_i 3 in Compare Match pulses

tunderflow_i 1 in Timer underflow

Figure 13: PWM generation unit



svenable_i 1 in Space Vector mode enable

svrdir_i 1 in Space Vector rotation direction

d_i 3 in Space Vector start condition

pwm_o 3 out Generated PWM

Table 6: PWM generation unit signals

 3.2.5 Dead-band time Unit

The Dead-Band time unit creates a clone of a PWM input signal and adds possibility to separate the 
edges of the PWM pair by a configurable time called dead-band time. By running one of the signals 
in the pair in opposite polarity any overlapping of active signal levels can be avoided. This is 
important for instance in a half-bridge where a PWM pair must not simultaneously have their 
corresponding transistors activated. Serving as time base is a counter register, DBTCNT, and when 
the Dead-Band unit is enabled this counter is started on either edge of the input PWM signal.

This implementation adds an offset of two sys_clk_i periods when activating a Dead-band time unit.

Dead-band time unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

pwm_i 1 in PWM

dbt_i 4 in Dead-band count range

dbtps_i 3 in Dead-band count prescaler setting

Figure 14: Dead-band time unit



edbt_i 1 in Enable Dead-band time operation

dtpwm_o1 1 out Dead-band time adjusted on PWM leading edge

dtpwm_o2 1 out Dead-band time adjusted on PWM trailing edge

Table 7: Dead-band time unit signals

 3.2.6 Full compare unit

The Full compare unit is responsible for PWM signal generation and is a composition of 3 Compare 
units, 1 PWM generation unit and 3 Dead-band time units. It is also possible to set the polarity of 
each PWM output from the Dead-band units as active high/low or forced high/low. This is 
controlled by the settings in the double buffered register Action-Control register, ACTR. Like the 
Compare units has ACTR configurable reload conditions.

The Full compare units support external Compare trip, ctrip signals which can serve as motor 
protection. Each Compare unit has its own ctrip connection and can be individually configured.

The Compare-Control register, COMCON register holds settings for:

• Enable compare operations

• Compare reload conditions

• ACTR reload conditions

• Enable Compare trip operations

The Dead-Band control register, DBTCON stores the configuration for the Dead-Band time units.

Full compare unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

data_i 16 in Data in

Figure 15: Full compare unit



reg_sel_i 3 in Register read/write selection

we_i 1 in Enable write to selected register

ctrip_i 3 in Compare trip events

cnt_i 16 in Current timer count

tpr_match_i 1 in Timer period match event

tunderflow_i 1 in Timer underflow event

data_o 16 out Data out from selected register

pwm_o 6 out PWM

cmpr_match_o 3 out Compare match events

Table 8: Full compare unit signals

 3.2.7 Edge detector

The purpose of the edge detectors is to synchronize an external signal to the clock domain of the 
Event Manager. This is done by a 3-bit shift register where if there is any difference between 2:nd 
and the 3:rd position that will be put out as a pulse on either positive edge, pos_o or negative edge, 
neg_o with pulse length of sys_clk_i period.

Edge Detector signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

sig_i 1 in Sampled external signal

pos_edge_o 1 out Pulse when positive edge detected

neg_edge_o 1 out Pulse when negative edge detected

Table 9: Edge Detector signals

 3.2.8 Capture unit

The capture units are based on a 2-level deep FIFO which is filled with the value of cnt_i on 
configurable external event and the purpose for this is usually to timestamp external events. The 
unit is designed to be used with a Edge Detector unit and the triggering methods can be configured 
by edge_trig_sel_i.

The FIFO can be read both from the FIFO-TOP and the FIFO-BOTTOM by settings of 
fifo_read_sel_i. To avoid flushing the FIFO on every sys_clk period the read_enable_i signal must 
be cycled between each read. FIFO status is given by fifo_status_o. Designed to be used with an 
external register, CAPCON which holds the configuration values of edge_trig_sel_i and 
fifo_read_sel_i.

capture unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

cnt_i 16 in Timer counting value

pos_edge_i 1 in Positive Edge pulse from edge detector

neg_edge_i 1 in Negative Edge pulse from edge detector

edge_trig_sel_i 2 in Select Capture trigger edge(s)

read_enable_i 1 in Enable read - FIFO update

fifo_read_sel_i 1 in FIFO read selection – top or bottom

fifo_status_o 2 out FIFO status

capture_o 16 out Capture value – depends on fifo_read_sel_i

capture_event_o 1 out Capture event pulse

Table 10: Capture unit signals

 3.2.9 Quadrature Encoder Pulse unit

In the quadrature-encoder-pulse, qep unit two equal pulse trains separated by a phase of 90 degrees 
is encoded to clock and direction signals. This is done by detecting a level change on one input and 
comparing that to the actual level on the other input a counting direction, qep_dir_o can be 
resolved. The qep_clk_o changes level on every edge detected on the inputs and thus gives a 
frequency twice of the input frequencies. This unit is designed to be used with two Edge Detectors.

Quadrature-Encoder-Pulse unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

pos_edge_i 2 in Positive Edge pulses from edge detectors

neg_edge_i 2 in Negative Edge pulses from edge detectors

qep_clk_o 1 out Generated qep clock

qep_dir_o 1 out Generated qep counting direction

Table 11: QEP unit signals

 3.2.10 Interrupt generator unit

The purpose of the Interrupt generator is to create level triggered CPU interrupt requests (IRQ) at 
events such as compare match or timer underflow. When a certain event pulse is detected the 

corresponding interrupt flag is set in either of the two flag registers IFRA or IFRB. Further on if the 
same interrupt event is unmasked in the mask registers IMRA or IMRB the corresponding interrupt 
line will be set for a CPU IRQ. An active interrupt is acknowledged by writing a logic one to 
corresponding flag.

Since there are over 20 different interrupts it might be desirable or needed to logically OR some or 
all interrupt lines. In that case, to be able to rapid determine active interrupt an interrupt 
identification register INTID was implemented. In INTID a unique number for active interrupt is 
presented which makes it possible to avoid search each of the flag registers when an IRQ happens. 
If two or more interrupts simultaneously happens the one with the highest priority will be presented 
by a predefined order.

Interrupt generator unit signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

sys_clk_i 1 in System clock

reset_i 1 in Asynchronous reset

ctrip_i 6 in Compare trip events

cmpr_match_i 6 in Compare match events

tpr_match_i 2 in Timer period match events

tunderflow_i 2 in Timer underflow events

toverflow_i 2 in Timer overflow events

capture_event_i 3 in Capture events

data_i 16 in Data in

reg_sel_i 3 in Register read/write selection

we_i 1 in Enable write to selected register

ctrip_int_o 6 out Compare trip interrupt request

cmpr_int_o 6 out Compare match interrupt request

tpr_int_o 2 out Timer period match interrupt request

tunderflow_int_o 2 out Timer underflow interrupt request

toverflow_int_o 2 out Timer overflow interrupt request

capture_event_int_o 3 out Capture event interrupt request

data_o 16 out Data out from selected register

Table 12: Interrupt generator unit signals

 3.3 Event Manager to Wishbone 8-bit interface

In order to connect the Event Manager to, for instance, the byte wide peripheral bus of the 
OpenRISC 1200 a 8-bit Wishbone interface was created. This implementation uses a 16-bit wide 
data latch register which contains either data to be written to or be read from the Event Manager and 
thus occupies 2 addresses. A third address is used for Event Manager register and read/write 
selection. Both clock and reset signals are used directly from the Wishbone bus by the Event 
Manager.

Register Name Address Offset Description

EV DATAHI 0x0 Event Manager data, high byte

EV DATALO 0x1 Event Manager data, low byte

EV REG 0x2 Event Manager Register selection and read/write operation
bit 5:0 : Event Manager register address
bit 7 : R/W op;  0 = read,  1 = write

Table 13: Event Manager 8-bit Wishbone interface Register map

To do a read operation one must first write the address of the desired Event Manager register to EV 
REG with bit 7 cleared. The Event Manager data will then be available to be read from EV 
DATAHI and EV DATALO.

The opposite procedure goes for write operation where data to the Event Manager first must be 
written to EV DATAHI and EV DATALO and then the address of Event Manager together with bit 
7 set are written to EV REG.

Event Manager 8-bit Wishbone interface signals

Signal Name Signal 
Width

Signal 
Direction

Signal Description

wb_clk 1 in Wishbone clock

wb_rst 1 in Wishbone reset

wb_adr_i 2 in Wishbone address

wb_dat_i 8 in Wishbone slave data in

wb_we_i 1 in Wishbone write enable

wb_cyc_i 1 in Wishbone bus cycle

wb_stb_i 1 in Wishbone strobe

wb_dat_o 8 out Wishbone slave data out

wb_ack_o 1 out Wishbone acknowledge

Event Manager specifics * in/out ctrips, irqs, external clocks, capture etc

Table 14: Event Manager Wishbone interface signals

 4 Results and discussion

 4.1 Development

Development was done using:

• Ubuntu Linux17 - Desktop Operating System

• Icarus Verilog18 - Verilog compiler and simulator

• GTKWave19 - Waveform viewer

• Microsemi Libero Suite20 - Full development suite for Actel FPGA:s. Integrated 
Design Environment (IDE), Compiler, Synthesizer, 
Placer & router, simulator and programmer

Although the Libero suite alone is sufficient for the FPGA development a decision was made to try 
out the open source tools Icarus Verilog and GTKWave to investigate how suitable these are for a 
project of this type.

The different modules of the Event Manager was implemented and simulated independently. When 
these behaved as intended the different parts were merged into the Event Manager together with the 
Wishbone interface.

Finally the total SoC build was created with the Event Manager IP core integrated to the OpenRISC 
1200 on its 8-bit peripheral bus and programmed to physical hardware on the μRTU development 
board.

During development each module was thoroughly tested by using Icarus Verilog simulator to reduce 
possible bugs. This turned out to be a success since after final build, using Libero suite to program 
the target hardware, very few corrections was required. This states that the open source tools used 
are a valid choice for HDL development on this level even though that commercial tools are 
required for the final product.

 4.2 Final testing

For testing, a simple bare-metal21 program for the OpenRISC and Event Manager was developed in 
C. This program receives commands by the UART22 via serial terminal on PC host. The commands 
consists of an address part and a data part targeting the Event Managers control registers. When an 
address followed by data is transmitted, a write to Event Manager is executed. If just an address is 
17 http://www.ubuntu.com
18 http://iverilog.icarus.com
19 http://gtkwave.sourceforge.net
20 http://www.actel.com/products/software/libero/default.aspx
21 A program that is executing directly on hardware without Operating System support
22 Universal Asynchronous Receiver Transmitter

transmitted a read from pointed register is attempted. This makes it easy to accomplish rapid test 
cases and can also be scripted on the PC side.

All functionality of the Event Manager had been thoroughly tested and bellow are principles and 
overview of the test cases presented:

GPTimers - For testing the timer functionality a setup with predefined timer period and 
compare values was used to generate a PWM signal with about 50% Duty 
Cycle (symmetric square wave). Then by using an oscilloscope or multimeter 
with frequency counting the timer frequency can be given. This was tested 
with different settings for timer period, counting mode and prescaler. Using 
clock triggering from the external pin and QEP was also tested.

PWM - By using the same analogy as with the timer but variate the compare value 
instead, thus giving different Duty Cycles, a percentage of the voltage of the 
PWM pin corresponding to actual Duty Cycle will be given on a multimeter. 
A faster method is off course to directly visualize the result on an 
oscilloscope. This was tested with different timing modes and different signal 
polarities. Deactivation of PWM on compare trip signals was also tested.

Capture - The quickest and most reliable way of testing the capture functionality is to 
make the capture triggering with the active timer disabled and with a 
predefined timer value. This and active timer mode was tested.

QEP - The quadrature-encoder pulse was tested by generating two PWM signals 
with square wave form and 90 degree phase difference and connecting these 
to the qep inputs. This was tested with different frequencies and lead signals 
(change of direction).

Interrupt unit - All interrupts, flags and masks plus interrupt identification functionality was 
tested. So was also the compare interrupt hold modules with different hold 
counts. A custom program was made for this which just notifies on UART 
when an active event occurred.

 4.3 Improvements

 4.3.1 Wishbone interface

The Wishbone interface presented in this report connects the clock of the Event Manager to the 
Wishbone clock and results in that the two are operated in the same clock domain. This is good 

since that avoids metastability. But in the other hand if a really fast PWM frequency is needed the 
Event Manager must be clocked by another faster clock and some form of domain bridge between 
the Wishbone interface and Event Manager is required.

Appendix A

A.1 Event Manager register map

EV REG Name Register Description

0x00 GPTCON GP Timer Control

0x01 T1CNT Timer 1 Counter value

0x03 T1PR Timer 1 Period value, TOP

0x04 T1CON Timer 1 Control

0x05 T2CNT Timer 2 Counter value

0x07 T2PR Timer 2 Period value, TOP

0x08 T2CON Timer 2 Control

0x11 COMCON1 Compare 1 Control

0x13 ACTR1 Compare Action Control 1

0x15 DBTCON1 Dead-Band Timer Control 1

0x17 CMPR1 Compare 1 value

0x18 CMPR2 Compare 2 value

0x19 CMPR3 Compare 3 value

0x1A COMCON2 Compare 2 Control

0x1B ACTR2 Compare Action Control 2

0x1C DBTCON2 Dead-Band Timer Control 2

0x1D CMPR4 Compare 4 value

0x1E CMPR5 Compare 5 value

0x1F CMPR6 Compare 6 value

0x20 CAPCON Capture Control

0x22 CAPFIFO Capture FIFO Status

0x23 CAP1FIFO Capture 1 FIFO TOP value

0x24 CAP2FIFO Capture 2 FIFO TOP value

0x25 CAP3FIFO Capture 3 FIFO TOP value

0x27 CAP1FBOT Capture 1 FIFO BOTTOM value

0x28 CAP2FBOT Capture 2 FIFO BOTTOM value

0x29 CAP3FBOT Capture 3 FIFO BOTTOM value

0x2C IMRA Interrupt Mask A

0x2D IMRB Interrupt Mask B

0x2F IFRA Interrupt Flag A

0x30 IFRB Interrupt Flag B

0x32 INTID Interrupt Identification value

0x34 CMPR1INTHLD Compare 1 Interrupt Hold value

0x35 CMPR2INTHLD Compare 2 Interrupt Hold value

0x36 CMPR3INTHLD Compare 3 Interrupt Hold value

0x37 CMPR4INTHLD Compare 4 Interrupt Hold value

0x38 CMPR5INTHLD Compare 5 Interrupt Hold value

0x39 CMPR6INTHLD Compare 6 Interrupt Hold value

0x3D ADCSOCHOLD ADC Start-Of-Conversion Hold value

0x3F VERSION Event Manager Version

Table 15: Register Address Map

Appendix B

B.1 Verilog Source code

B.1.1 Full Compare unit
// ev_fullcmpr.v 

// Module containing 3 compare units, one PWM signal generator with space vector 

// functionality and deadband units for each PWM signal. 

/* 

 * Signals 

 * 

 * data_i: data in for loading selected register 

 * cnt_i: timer count in 

 * pr_match_i: timer period match signal in 

 * tunderflow_i: timer underflow signal in 

 * reg_sel_i: internal register selection in 

 * 'h1: CMPR1 

 * 'h2: CMPR2 

 * 'h3: CMPR3 

 * 'h4: COMCON 

 * 'h5: ACTR 

 * 'h6: DBTCON 

 * we_i: enable write on selected register 

 * ctrip_i: compare trips in 

 * sys_clk_i: clock source 

 * reset_i: reset in - asynchronous 

 * data_o: data out from selected register 

 * pwm_o: PWM outputs 

 * cmpr_match_o: compare match out signals 

 */ 

`include "ev_defines.v" 

module ev_fullcmpr ( 

data_i, 

cnt_i, 

pr_match_i, 

tunderflow_i, 

reg_sel_i, 

we_i, 

ctrip_i, 

sys_clk_i, 

reset_i, 

data_o, 

pwm_o, 

cmpr_match_o 

); 

parameter DATA_WIDTH = `EV_REG_WIDTH; 

// Interface declarations ------------------------------------------- 

input [DATA_WIDTH-1:0] data_i, cnt_i; 

input pr_match_i, tunderflow_i, we_i, sys_clk_i, reset_i; 

input [2:0] reg_sel_i; 

input [3:1] ctrip_i; 

 

output [DATA_WIDTH-1:0] data_o; 

output [6:1] pwm_o; 

output [3:1] cmpr_match_o; 

 

 

// Register declarations --------------------------------------------- 

reg [`EV_CENABLE:`EV_C1TRIPE] comcon; 

reg [DATA_WIDTH-1:0] actr, actr_shdw; 

reg [`EV_DBT3:`EV_DBTPS0] dbtcon; 

reg actr_reload; 

// Connections ------------------------------------------------------- 

wire [DATA_WIDTH-1:0] cmpr_out[1:3]; 

wire [3:1] pwm; 

wire [6:1] dtpwm; 

// -------------------------------------------------------------------

// Module instantiations

//--------------------------------------------------------------------

// compare units ----------------------------------------------------- 

ev_cmpr #(DATA_WIDTH) CMPR_U1 (  

.cmpr_i(data_i), 

.cnt_i(cnt_i), 

.cld_i(comcon[`EV_CLD1:`EV_CLD0]), 

.ecmpr_i(comcon[`EV_CENABLE]), 

.we_i((reg_sel_i == 3'h1) & we_i), 

.pr_match_i(pr_match_i), 

.underflow_i(tunderflow_i), 

.cnt_dir_i(1'b1), // not used here 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.cmpr_o(cmpr_out[1]), 

.cmpr_match_o(cmpr_match_o[1]), 

.pwm_o() // not used here 

); 

 

ev_cmpr #(DATA_WIDTH) CMPR_U2 (  

.cmpr_i(data_i), 

.cnt_i(cnt_i), 

.cld_i(comcon[`EV_CLD1:`EV_CLD0]), 

.ecmpr_i(comcon[`EV_CENABLE]), 

.we_i((reg_sel_i == 3'h2) & we_i), 

.pr_match_i(pr_match_i), 

.underflow_i(tunderflow_i), 

.cnt_dir_i(1'b1), // not used here 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.cmpr_o(cmpr_out[2]), 

.cmpr_match_o(cmpr_match_o[2]), 

.pwm_o() // not used here 

); 

 

ev_cmpr #(DATA_WIDTH) CMPR_U3 (  

.cmpr_i(data_i), 

.cnt_i(cnt_i), 

.cld_i(comcon[`EV_CLD1:`EV_CLD0]), 

.ecmpr_i(comcon[`EV_CENABLE]), 

.we_i((reg_sel_i == 3'h3) & we_i), 

.pr_match_i(pr_match_i), 

.underflow_i(tunderflow_i), 

.cnt_dir_i(1'b1), // not used here 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.cmpr_o(cmpr_out[3]), 

.cmpr_match_o(cmpr_match_o[3]), 

.pwm_o() // not used here 

); 

// ------------------------------------------------------------------- 

 

 

// PWM & Space vector unit -------------------------------------------  

ev_sv_pwm SV_PWM ( 

.cmpr_match_i(cmpr_match_o), 

.tunderflow_i(tunderflow_i), 

.svenable_i(comcon[`EV_SVENABLE]), 

.svrdir_i(actr[`EV_SVRDIR]), 

.d_i(actr[`EV_D2:`EV_D0]), 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.pwm_o(pwm) 

); 

// ------------------------------------------------------------------- 

 

 

// Dead-band-time units ----------------------------------------------  

ev_dbtu DBU1 ( 

.pwm_i(pwm[1]), 

.dbt_i(dbtcon[`EV_DBT3:`EV_DBT0]), 

.edbt_i(dbtcon[`EV_EDBT1]), 

.dbtps_i(dbtcon[`EV_DBTPS2:`EV_DBTPS0]), 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.dtpwm_o1(dtpwm[1]), 

.dtpwm_o2(dtpwm[2]) 

); 

 

ev_dbtu DBU2 ( 

.pwm_i(pwm[2]), 

.dbt_i(dbtcon[`EV_DBT3:`EV_DBT0]), 

.edbt_i(dbtcon[`EV_EDBT2]), 

.dbtps_i(dbtcon[`EV_DBTPS2:`EV_DBTPS0]), 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.dtpwm_o1(dtpwm[3]), 

.dtpwm_o2(dtpwm[4]) 

); 

 

ev_dbtu DBU3 ( 

.pwm_i(pwm[3]), 

.dbt_i(dbtcon[`EV_DBT3:`EV_DBT0]), 

.edbt_i(dbtcon[`EV_EDBT3]), 

.dbtps_i(dbtcon[`EV_DBTPS2:`EV_DBTPS0]), 

.sys_clk_i(sys_clk_i), 

.reset_i(reset_i), 

.dtpwm_o1(dtpwm[5]), 

.dtpwm_o2(dtpwm[6]) 

); 

// ------------------------------------------------------------------- 

// Procedure ---------------------------------------------------------

always @ (posedge sys_clk_i or posedge reset_i) begin 

if (reset_i) begin 

actr <= 0; 

actr_shdw <= 0; 

comcon[`EV_CENABLE:`EV_FCMP1OE] <= 0; 

comcon[`EV_C3TRIPE:`EV_C1TRIPE] <= 3'h7; //ctrips enabled by default 

dbtcon <= 0; 

end else if (sys_clk_i) begin 

if (we_i) begin 

case (reg_sel_i) 

3'h4: 

comcon <= data_i[`EV_CENABLE:`EV_C1TRIPE]; 

3'h5: 

actr_shdw <= data_i; 

 

3'h6: 

dbtcon <= data_i[`EV_DBT3:`EV_DBTPS0]; 

endcase 

end 

// reload condition? 

if (actr_reload) 

actr <= actr_shdw; 

 

// disable outputs if 'ctrip'-event 

if (comcon[`EV_C1TRIPE] & ctrip_i[1]) 

comcon[`EV_FCMP1OE] <= 1'b0; 

if (comcon[`EV_C2TRIPE] & ctrip_i[2]) 

comcon[`EV_FCMP2OE] <= 1'b0; 

if (comcon[`EV_C3TRIPE] & ctrip_i[3]) 

comcon[`EV_FCMP3OE] <= 1'b0; 

 

end 

end 

 

// comb. process for actr_reload 

always @ ( comcon[`EV_ACTRLD1:`EV_ACTRLD0] or pr_match_i or 

tunderflow_i or comcon[`EV_CENABLE]) begin 

if (~comcon[`EV_CENABLE]) 

actr_reload <= 1'b1; 

else case (comcon[`EV_ACTRLD1:`EV_ACTRLD0]) 

2'b00: 

if (tunderflow_i) 

actr_reload <= 1'b1; 

else 

actr_reload <= 1'b0; 

 

2'b01: 

if (tunderflow_i | pr_match_i) 

actr_reload <= 1'b1; 

else 

actr_reload <= 1'b0; 

 

2'b10: 

actr_reload <= 1'b1; 

 

default: 

actr_reload <= 1'b0; 

endcase 

end 

 

// data output depending on 'reg_sel_i' 

assign data_o = (reg_sel_i == 3'h1) ? cmpr_out[1] : 

(reg_sel_i == 3'h2) ? cmpr_out[2] : 

(reg_sel_i == 3'h3) ? cmpr_out[3] : 

(reg_sel_i == 3'h4) ? {comcon, 4'h0} : 

(reg_sel_i == 3'h5) ? actr : 

(reg_sel_i == 3'h6) ? dbtcon : 

{DATA_WIDTH{1'b0}}; 

 

// PWM output assignments 

assign pwm_o[1] = comcon[`EV_FCMP1OE] & (~comcon[`EV_C1TRIPE] | ~ctrip_i[1]) ? 

(actr[`EV_CMP1ACT1:`EV_CMP1ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP1ACT1:`EV_CMP1ACT0] == 2'b01) ? ~dtpwm[1] : 

(actr[`EV_CMP1ACT1:`EV_CMP1ACT0] == 2'b10) ? dtpwm[1] : 1'b1 : 

1'bz; 

assign pwm_o[2] = comcon[`EV_FCMP1OE] & (~comcon[`EV_C1TRIPE] | ~ctrip_i[1]) ? 

(actr[`EV_CMP2ACT1:`EV_CMP2ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP2ACT1:`EV_CMP2ACT0] == 2'b01) ? ~dtpwm[2] : 

(actr[`EV_CMP2ACT1:`EV_CMP2ACT0] == 2'b10) ? dtpwm[2] : 1'b1 : 

1'bz; 

assign pwm_o[3] = comcon[`EV_FCMP2OE] & (~comcon[`EV_C2TRIPE] | ~ctrip_i[2]) ? 

(actr[`EV_CMP3ACT1:`EV_CMP3ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP3ACT1:`EV_CMP3ACT0] == 2'b01) ? ~dtpwm[3] : 

(actr[`EV_CMP3ACT1:`EV_CMP3ACT0] == 2'b10) ? dtpwm[3] : 1'b1 : 

1'bz; 

assign pwm_o[4] = comcon[`EV_FCMP2OE] & (~comcon[`EV_C2TRIPE] | ~ctrip_i[2]) ? 

(actr[`EV_CMP4ACT1:`EV_CMP4ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP4ACT1:`EV_CMP4ACT0] == 2'b01) ? ~dtpwm[4] : 

(actr[`EV_CMP4ACT1:`EV_CMP4ACT0] == 2'b10) ? dtpwm[4] : 1'b1 : 

1'bz; 

assign pwm_o[5] = comcon[`EV_FCMP3OE] & (~comcon[`EV_C3TRIPE] | ~ctrip_i[3]) ? 

(actr[`EV_CMP5ACT1:`EV_CMP5ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP5ACT1:`EV_CMP5ACT0] == 2'b01) ? ~dtpwm[5] : 

(actr[`EV_CMP5ACT1:`EV_CMP5ACT0] == 2'b10) ? dtpwm[5] : 1'b1 : 

1'bz; 

assign pwm_o[6] = comcon[`EV_FCMP3OE] & (~comcon[`EV_C3TRIPE] | ~ctrip_i[3]) ? 

(actr[`EV_CMP6ACT1:`EV_CMP6ACT0] == 2'b00) ? 1'b0 : 

(actr[`EV_CMP6ACT1:`EV_CMP6ACT0] == 2'b01) ? ~dtpwm[6] : 

(actr[`EV_CMP6ACT1:`EV_CMP6ACT0] == 2'b10) ? dtpwm[6] : 1'b1 : 

1'bz; 

endmodule 

B.1.2 PWM Generation unit

// ev_sv_pwm 

// Module for generating PWM signals from compare match events. 

// uses compare match signals from 3 compare units and timer 

// underflow signal. 

// Space vector functionality based on TMS320x281x is implemented, please 

// consult manual for TMS320x281x for futher info 

/* 

 * Signals 

 * 

 * cmpr_match_i: compare match signals in 

 * tunderflow_i: timer underflow signal in 

 * svenable_i: enable space vector pwm 

 * svrdir_i: space vector rotation direction 

 * d_i: space vector start condition 

 * sys_clk_i: clock source 

 * reset_i: reset in - asyncronous 

 * pwm_o: pwm out 

 * 

 */ 

 

`include "ev_defines.v" 

module ev_sv_pwm ( 

cmpr_match_i, 

tunderflow_i, 

svenable_i, 

svrdir_i, 

d_i, 

sys_clk_i, 

reset_i, 

pwm_o 

); 

 

input [3:1] cmpr_match_i; 

input tunderflow_i, svenable_i, svrdir_i; 

input [2:0] d_i; 

input sys_clk_i, reset_i; 

 

output [3:1] pwm_o; 

 

 

reg [3:1] pwm, save_sv_state; 

 

 

always @ (posedge sys_clk_i or posedge reset_i) begin 

if (reset_i) begin 

pwm <= 0; 

save_sv_state <= 0; 

end else if (sys_clk_i) begin 

if (svenable_i) begin // use space vector pwm? 

if (tunderflow_i) 

pwm <= d_i; 

else if (cmpr_match_i[1]) begin 

if (pwm == d_i) begin 

case(d_i) 

'b000, 'b111: begin 

pwm <= d_i; 

save_sv_state <= d_i;

end 

 

'b001: 

if (svrdir_i) 

pwm <= 'b101; 

else 

pwm <= 'b011; 

'b010: 

if (svrdir_i) 

pwm <= 'b011; 

else 

pwm <= 'b110; 

'b011: 

if (svrdir_i) 

pwm <= 'b001; 

else 

pwm <= 'b010; 

'b100: 

if (svrdir_i) 

pwm <= 'b110; 

else 

pwm <= 'b101; 

'b101: 

if (svrdir_i) 

pwm <= 'b100; 

else 

pwm <= 'b001; 

'b110: 

if (svrdir_i) 

pwm <= 'b010; 

else 

pwm <= 'b100; 

endcase 

end else 

pwm <= d_i; 

end else if (cmpr_match_i[2]) begin 

case (pwm) 

'b000, 'b111: 

pwm <= save_sv_state; 

'b001, 'b010, 'b100: begin 

save_sv_state <= pwm; 

pwm <= 'b000; 

end 

'b011, 'b101, 'b110: begin 

save_sv_state <= pwm; 

pwm <= 'b111; 

end 

endcase 

end 

end else begin // generate "normal" PWM 

pwm <= ( 

(pwm ^ cmpr_match_i) & {3{~tunderflow_i}}) | 

(cmpr_match_i & {3{tunderflow_i}} 

); 

end 

end 

end 

 

assign pwm_o = pwm;  

endmodule 

